ASP-BOE – February 19, 2011

“ASP-BOE Fast Data” – v.1
' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-------------------------------------------------

ShtDta  PIN 1        ' bi-directional data with pullup resistor

ShtClk  PIN 0        ' clock (for Sensirion) with pulldown resistor

DSclk    PIN 2      ' DS1302 RTC clock

DSdta    PIN 5      ' DS1302 RTC data line

DSce    PIN 7      ' DS1302 RTC chip enable (high enable)

' note p3, p4, and p6 are unused

TX     PIN 8               ' Transmit Data   --> 27937.4 (RXD) output to Vinculum

RTS    PIN 9               ' Request To Send --> 27937.6 (CTS) to Vinculum

RX    PIN 10              ' Receive Data    <-- 27937.5 (TXD) from Vinculum

CTS    PIN 11              ' Clear To Send   <-- 27937.2 (RTS) from Vinculum

ledP    PIN 12      ' indicator LED, positive side

ledN    PIN 13      ' indicator LED, negative side

' -----[ Constants ]-------------------------------------------------------

vBaud           CON     84    ' Serial Baud Rate 9600 bps for Vinculum datalogger

ShtTemp         CON     %00011    ' read temperature command for Sensirion

ShtHumi         CON     %00101    ' read humidity command for Sensirion

Ack             CON     0      ' responses for Sensirion

NoAck           CON     1

DegSym          CON     186     ' degrees symbol for DEBUG

' -----[ Variables ]-------------------------------------------------------

primo         VAR Word

  iobyte  VAR primo.BYTE0    ' io to/from datalogger, humidty sensor, & debug

  index  VAR primo.NIB2    ' loop counter in some routines

  unused1  VAR primo.NIB3

secundo  VAR Word

  flags  VAR secundo.BYTE0

   ackbit  VAR flags.BIT0    ' acknowledge Sensirion

   shtErr   VAR flags.BIT1    ' = 1 if error in sensirion read

   dskErr  VAR flags.BIT2    ' =1 if the logger fails to work

   land       VAR flags.BIT3       ' =1 after alloted time interval

  unused2  VAR secundo.BYTE1

waldo         VAR Word             ' scratch variable

  soSHT  VAR waldo      ' alias as scratch variable for Sensirion

  soSHT0  VAR soSHT.BYTE0    ' byte by byte

  soSHT1  VAR soSHT.BYTE1

  delay  VAR waldo      ' timeout for Sensirion

  ' motormovementtime  VAR waldo  ' alias as timer variable for movements

degC          VAR Word             ' temperature in Celsius*10 from Sensirion

rh            VAR Word             ' humidity in percent*10 from Sensirion

time0         VAR Word

  second  VAR time0.BYTE0    ' from DS1302 RTC

  minute  VAR time0.BYTE1    ' ditto

time1      VAR Word

  hour    VAR time1.BYTE0    ' ditto

  index2    VAR time1.BYTE1    ' extra index for loop count

'buffer   VAR Word                  ' for data returned from logger or RTC

'  buf0  VAR buffer.BYTE0  ' will return as bytes

'  buf1  VAR buffer.BYTE1

array   VAR Word

  alt0  VAR array.BYTE0

  alt1  VAR array.BYTE1

' -----[ Initialization ]--------------------------------------------------

Reset:

  OUTS =  %0001001000000000   ' indicator led is on. rts (p9) high to start, tx (p8) low to start.

  'fedcba9876543210

  DIRS =  %1111001111111101   ' p1 is input for sensirion data, pa and pb input for rx and cts

  DEBUG CR, CR, "Connecting circuit board and flash drive..."

  DEBUG CR, CR, "When they are connected and synchronized it will"

  DEBUG CR,     "measure and record temperature, humidity, and time", CR, CR

  DEBUG "____________________________________________________ ", CR

  PAUSE 10

  GOSUB RTC_init  ' reset the clock to 00:00:00

  PAUSE 1000

  GOSUB RTC_init '----- mk added 08 March 2008-------

  PAUSE 512

  HIGH TX                               ' Initialize Transmit Line

  LOW RTS                               ' Take Vinculum Out Of Reset

  PAUSE 512                               ' Allow 1/2 Second To Settle

  LOW ledP        ' LED indicator off - long pulse signals reset

  GOSUB V_synchronize           ' get ready to send data to thumb drive

  DEBUG CR,"sync tries: ",DEC waldo

  iobyte=waldo : GOSUB morse_numeral   ' show the tries count for debugging

  GOSUB V_short_commands  ' enable short command mode on Vinculum/thumb drive

  GOSUB V_Check_Drive

  DEBUG CR,"disk tries: ",DEC waldo, CR

  iobyte=waldo : GOSUB morse_numeral   ' show the tries count for debugging

  ' either print disk not found, else write

  IF dskerr=1 THEN

    DEBUG " not found"

   ELSE

    GOSUB V_open_datafile

    SEROUT TX\CTS, vBaud, [ $8,$20,0,0,0,11,CR,"SHT data",CR, LF, CR]

    GOSUB V_purge

    GOSUB V_close_datafile

   ENDIF

  DEBUG CR,"Sample time   Sample#   Temp(C) Humidity(%)",BELL,CR

  GOSUB SHT_Connection_Reset

  PAUSE 256

' -----[ Program Code ]----------------------------------------------------

Main:

  DO

    PULSOUT ledP,50

    GOSUB RTC_read

' mk removed " minute =$01 OR " from the next line, right after IF command - 01 July 2009

    IF index2>100 THEN land=1   ' set flag to start movements after descent of one minute

    index2 = index2 + 1

    DEBUG CR,HEX2 hour,":",HEX2 minute,":",HEX2 second, TAB, DEC index2

    GOSUB SHT_Measure_Temp

    GOSUB SHT_Measure_Humidity

    DEBUG TAB,REP "-"\degC.BIT15,DEC ABS degC/10,".",DEC1 ABS degC,degSym

    DEBUG TAB, DEC rH/10, ".", DEC1 rH, "%", CR

    GOSUB V_open_datafile  ' thumb drive

    GOSUB V_write_record   ' one data record

    GOSUB V_close_datafile  ' close the file

'    IF land THEN GOSUB movement - mk commented out - 01 July 2009

  LOOP

  END

' -----[ Subroutines ]-----------------------------------------------------

' -----[ RTC ]-----

RTC_init:

  ' send $00 to DS1302 control register (command $8E) to enable write

  HIGH DSce

  SHIFTOUT DSdta,DSclk,LSBFIRST,[$8E,$00]

  LOW DSce

  ' set the clock time using burst command $BE

  ' second=0, minute=0, hour=0, date31=2, month=3, day7=1, year=08

  ' also disable further writes: $80 at loc $7

  HIGH DSce

  SHIFTOUT DSdta,DSclk,LSBFIRST,[$BE,$00,$00,$00,$02,$03,$01,$08,$80]

  LOW DSce

RETURN

RTC_read:

  ' retrieves the time to variable: second, minute, hour

  HIGH DSce

    SHIFTOUT DSdta,DSclk,LSBFIRST,[$BF]     ' burst mode read $BF command

    SHIFTIN DSdta,DSclk,LSBPRE,[second,minute,hour]

  LOW DSce

  RETURN

' -----[ Vinculum/Thumb Drive]-----

' refer to email discussion around email http://forums.parallax.com/forums/default.aspx?f=6&p=7&m=247882#m252750

V_purge:

V_Get_Data:

  SERIN RX\RTS, vbaud, 16, V_Timeout, [STR alt0\12]  'number after vbaud controls data read-write speed

  'SERIN 15,84,20,escape,[STR array\7]  'the last of the 3 double digits controls data read-write speed

                                       ' and also controls Datalogger boot-up rate

 ' IF array(6)=LF THEN DEBUG STR array\7   ' LF in array(6) means good data.

escape:

V_Timeout:  ' timeout is the normal exit, buffer not full

  RETURN

V_synchronize:

'  Send 'E' to start the synchronization process.

'  Expect 'E' and a CR for the response, timing per pk

waldo=0   ' count tries

DO

  GOSUB V_purge   ' clear the vinculum tx data buffer

  PAUSE 256     ' wait about 1/4 second

  SEROUT TX\CTS, vbaud, ["E", CR]        ' Sync Command Character

  GOSUB V_get_Data                        ' Get Response

  waldo=waldo+1

LOOP UNTIL (alt0="E" AND alt1=CR) OR waldo=8

  GOSUB V_purge                     ' Purge Buffer of leftover junk

  SEROUT TX\CTS, vbaud, ["e", CR]        ' next Sync Command Character

  GOSUB V_get_Data                        ' Get Response/purge

  IF (alt0<>"e" OR alt1<>CR) THEN waldo=waldo + 1

  IF waldo>7 THEN dskerr=1

  ' returns with number of tries in waldo

  ' returns with dskerr=1 if cant sync

RETURN

V_Check_Drive:

  waldo=0

  DO

    GOSUB V_purge

    SEROUT TX\CTS, vbaud, [CR]

    GOSUB V_get_Data

    IF alt0="N" AND alt1="D" THEN

      waldo=9   ' error: disk not found

      EXIT     ' no point to test more

    ENDIF

    waldo=waldo+1

    PAUSE 100

  LOOP UNTIL alt0=">" OR waldo=8

  IF waldo>7 THEN dskerr=1

  ' return with number of tries in waldo

  ' waldo=9 means it got the no disk error

  ' dskerr flag=1 is tries>7

RETURN

V_short_commands:

  SEROUT TX\CTS, vbaud, ["SCS", CR]

  GOSUB v_get_Data                       ' purge

  RETURN

V_Open_DataFile:

  GOSUB V_purge      ' purge receive buffer

  SEROUT TX\CTS, vbaud, [$09, $20, "datafile.txt", CR]

  GOSUB V_purge      ' purge receive buffer

  RETURN

v_Close_DataFile:

  GOSUB V_purge                      ' Purge Receive Buffer

  SEROUT TX\CTS, vbaud, [$0A, $20, "datafile.txt", CR]

  GOSUB V_purge             ' purge receive buffer

  RETURN

V_write_record:

' Write data with time to flash drive

 SEROUT TX\CTS, vBaud, [ $8,$20,0,0,0,24,CR, HEX2 hour, ":" , HEX2 minute, ":", HEX2 second, "," , degC.BIT15*13+32,DEC3 ABS (degC/10),".", DEC1 ABS degC,",",DEC3 (RH/10),".",DEC1 RH,CR, LF, CR]

 GOSUB V_purge

RETURN

' -----[ Sensirion ]-----

SHT_Connection_Reset:

  SHIFTOUT ShtDta, ShtClk, LSBFIRST, [$FFF\9]

  RETURN

'

' generates SHT1x "start" sequence

'          _____         _____

' ShtData       |_______|

'              ___     ___

' Clock    ___|   |___|   |___

SHT_Start:

  INPUT ShtDta    ' let pull-up take line high

  LOW ShtClk

  HIGH ShtClk

  LOW ShtDta

  LOW ShtClk

  HIGH ShtClk

  INPUT ShtDta

  LOW ShtClk

  RETURN

' sends start sequence & command byte & retrieves two data bytes from Sensirion

SHT_measure:

  GOSUB SHT_Start                               ' alert device

  GOSUB SHT_Write_Byte                          ' send command (iobyte)

  GOSUB SHT_Wait                                ' wait until measurement done

  ackBit = Ack                                  ' another read follows

  GOSUB SHT_Read_Byte                           ' get MSB

  soSHT1 = ioByte

  ackBit = NoAck                                ' last read

  GOSUB SHT_Read_Byte                           ' get LSB

  soSHT0 = ioByte

  RETURN

'

' measure temperature:

' -- celsius = soT * 0.01 - 40

SHT_Measure_Temp:

  ioByte = ShtTemp                             ' temperature command

  GOSUB SHT_measure                         ' send command

  degC = soSHT / 10 - 400                           ' convert to tenths C

  RETURN

' measure humidity

SHT_Measure_Humidity:

  ioByte = ShtHumi                             ' humidity command

  GOSUB SHT_measure                             ' send command

  rh = (soSHT ** 26542)

  rh = rh - ((soSHT ** 3468) * (soSHT ** 3468) + 50 / 100)

  rh = rh - 40

  rh = ((degC + 250 / 10 - 50) * (soSHT ** 524 + 1) + (rh * 10)) + 5 / 10

  RETURN

' sends "ioByte"

' returns "ackBit"

SHT_Write_Byte:

  SHIFTOUT ShtDta, ShtClk, MSBFIRST, [ioByte]  ' send byte

  SHIFTIN  ShtDta, ShtClk, LSBPRE, [ackBit\1]   ' get ack bit

  RETURN

' returns "ioByte"

' sends "ackBit"

SHT_Read_Byte:

  SHIFTIN  ShtDta, ShtClk, MSBPRE, [ioByte]    ' get byte

  SHIFTOUT ShtDta, ShtClk, LSBFIRST, [ackBit\1] ' send ack bit

  INPUT ShtDta                                 ' release data line

  RETURN

' wait for sht to finish measurement

' (data line pulled low by SHT within 1/4 second)

SHT_Wait:

  INPUT ShtDta                                 ' data line is input

  delay = 255                        ' give >1/4 second to finish

  DO

  delay=delay-1

  PAUSE 1

  LOOP UNTIL ShtDta=0 OR delay=0

  IF delay=0 THEN

    GOSUB SHT_Connection_Reset

    shtErr=1

  ENDIF

  RETURN

' -----[ morse code ]-----

Morse_numeral:

iobyte = %11110000011111 >> iobyte & %11111 | %10100000

Morse:

FOR index = iobyte>>5 TO 1

  HIGH ledP

  PAUSE 192*iobyte.BIT0(index-1)+64

  LOW ledP

  PAUSE 256

  NEXT

  PAUSE 512

  RETURN

morse_reading:

DEBUG 32

  FOR ackbit=1 TO 0

  iobyte = waldo DIG ackbit

  DEBUG DEC1 iobyte

  GOSUB morse_numeral

  NEXT

  RETURN

mcs CON 3*32+%000  ' dit dit dit

mco CON 3*32+%111 ' dah dah dah

mcn CON 2*32+%10   ' dah dit

mcf CON 4*32+%0010  ' dit dit dah dit

mcb CON 4*32+%1000   ' dah dit dit dit

mcc CON 4*32+%1010  ' dah dit dah dit

mcw CON 3*32+%011  ' dit dah dah

mce CON 1*32+%0     ' dit

mcv CON 4*32+%0001  ' dit dit dit dah

morse_message:

' plays back morse message up to null

' enter with waldo pointing to message start

' e.g. waldo=message1 : GOSUB morse_message

' message1 DATA mcs,mco,mcs,0  ' sos, example message for LED

DO

  READ waldo,iobyte

  IF iobyte=0 THEN EXIT

  GOSUB morse

  waldo=waldo+1

LOOP

  RETURN

' -----[ servos ]-----

'movement:

'   PAUSE 100

'   iobyte=mcf : GOSUB morse

'  GOSUB forward

'   iobyte=mcb : GOSUB morse

'  GOSUB backward

'   iobyte=mcc : GOSUB morse

'  GOSUB turnclockwise

'   iobyte=mcf : GOSUB morse

'  GOSUB forward

'   iobyte=mcb : GOSUB morse

'  GOSUB backward

'   iobyte=mcw : GOSUB morse

'  GOSUB turnCOUNTERclockwise

'   iobyte=mce : GOSUB morse

'    PAUSE 1000

'  RETURN

'forward:

'   FOR motormovementtime = 0 TO 120

'   PULSOUT servoL, 850                           ' move forward 1

'   PULSOUT servoR, 650

'   PAUSE 20

'   NEXT

'   RETURN

'backward:                                    ' move backward 1/3

'   FOR motormovementtime = 0 TO 60

'   PULSOUT servoL, 650

'   PULSOUT servoR, 850

'   PAUSE 20

'   NEXT

'   RETURN

'turnclockwise:                               ' rotate clockwise

'   FOR motormovementtime = 0 TO 30

'   PULSOUT servoL, 850

'   PULSOUT servoR, 850

'   PAUSE 20

'   NEXT

'   RETURN

'turnCOUNTERclockwise:                               ' rotate clockwise

'   FOR motormovementtime = 0 TO 30

'   PULSOUT servoR, 650

'   PULSOUT servoL, 650

'   PAUSE 20

'   NEXT

'   RETURN
